



СОЗДАНИЕ ВНЕШНЕЙ ПЕЧАТНОЙ ФОРМЫ С НУЛЯ

В этом уроке мы будем самостоятельно делать внешнюю печатную форму с нуля без конструктора печати.

Создаем новый файл. Внешняя обработка. Называем ее как-то, например, внешнее коммерческое предложение.



Нажимаем макеты, добавить.



Оставим название макета, пусть он называется макет. Менять его не будем. Нажимаем готово.

Первым делом нужно решить на сколько областей поделить наш документ.

Например Шапка, ТоварыШапка, Товары, Подвал.

Вот так нам нарисовал конструктор запроса, почему он решил поделить этот документ на области? Здесь все дело в том, что в этой печатной форме нам нужно выводить данные из табличной части.

И мы с вами знаем, что доступ к данным табличной части мы можем получить только в цикле, перебирая эту табличную часть.

Поэтому принято делать так, что под данные строки табличной части заводится отдельная область, и эта область выводится в цикле столько раз, сколько строк в табличной части.

Поэтому нам для вывода строк табличной части нужна отдельная область. И вот уже получается, что у нас как минимум на три области шапка, строка табличной части и подвал, разбивается весь наш документ. И плюс еще шапка табличной части. Тоже обычно для нее делается отдельная область.

Вывод областей регулируется кодом. Как мы в коде напишем, в таком порядке области и выведутся. Поэтому - основа нашего макета – это области.

Давайте пойдём и будем их создавать.

Мы можем взять и выделить несколько строк, нажать «Таблица», «Имена», назначить имя и назначить имя этой области.

Какой ширины примерно делать печатную форму, чтобы она была на лист А4? Считается, что это примерно 12 колонок.

Теперь мы оформляем наш макет.

Мы пишем весь текст, который в нем должен быть, и вставляем все параметры, которые хотели бы в нем увидеть. Я рекомендую вам - называть параметры точно также как называются реквизиты документа.

Шапка	1												
	2												
	3												
	4												
	5												
	6												
	7												
	8												
	9												
	10												
	11												
Товары	12	№	Наименование	Кол-во	Цена	Сумма							
Товары	13												
	14	<НомерСтр>	<Номенклатура>	<Количество>	<Цена>	<Сумма>							
	15												
Подвал	16												
	17	Всего:	<СуммаДокумента>	<СуммаДокумента>	Прописью>								
	18												
	19												
	20												
	21												
	22	Генеральный директор		<Руководитель>									
	23												
	24												
	25												
	26												
	27												

Чтобы этот макет заполнялся, нам нужен код. Где располагается код печатной формы? Он располагается в ее модуле объекта.

Действия / Открыть модуль объекта. Вставляете в модуль объекта код, который я вам дам.

Прикреплю к уроку код модуля объекта печатной формы и он всегда будет одинаковым в каждой конфигурации, в каждой внешней печатной форме.

Мы его весь копируем и весь вставляем просто в модуль нашего объекта. Здесь мы меняем только то название документа, для которого предназначена ваша внешняя печатная форма. Дальше у вас все остается так же, как было, и вы должны прописать заполнение параметров макета в функции «Сформировать печатную форму».

С чего начинается получение любых данных? Конечно же с запроса. Чтобы заполнить параметры, нам нужно написать запрос.

Конструктор запроса с обработкой результата. Создать новый запрос. Переходим на таблицы и поля, выбираем нужный документ. Счет на оплату покупателю. Во-первых, вытаскиваем в список таблиц тот документ, для которого мы делаем внешнюю печатную форму.

Дальше смотрим, что нам нужно вывести. Итак, нам нужно взять организацию, ее ИНН и КПП, дату. Нам нужна сумма документа и руководитель.

Теперь нам нужна табличная часть. А, точнее, нам нужны ее колоночки. Номер, наименование, количество, цена, сумма. Когда мы делаем внешнюю печатную форму, нам не нужно присоединять

отдельно табличную часть, мы можем прямо здесь выбрать нужные нам реквизиты табличной части. Делаем это. Номер строки, номенклатура, количество, цена, сумма.

И дальше нам еще нужно отобразить по ссылке. Нажимаем ОК. У нас появляется запрос.

```
Для Каждого Ссылка Из МассивОбъектов Цикл
// ЗДЕСЬ НАПИСАТЬ ЗАПОЛНЕНИЕ ПАРАМЕТРОВ МАКЕТА

Запрос = Новый Запрос;
Запрос.Текст =
"ВЫБРАТЬ
|   СчетНаОплатуПокупателю.Организация КАК Организация,
|   СчетНаОплатуПокупателю.Организация.ИНН КАК ОрганизацияИНН,
|   СчетНаОплатуПокупателю.Организация.КПП КАК ОрганизацияКПП,
|   СчетНаОплатуПокупателю.Дата КАК Дата,
|   СчетНаОплатуПокупателю.СуммаДокумента КАК СуммаДокумента,
|   СчетНаОплатуПокупателю.Руководитель КАК Руководитель,
|   СчетНаОплатуПокупателю.Товары. (
|       НомерСтроки КАК НомерСтроки,
|       Номенклатура КАК Номенклатура,
|       Количество КАК Количество,
|       Цена КАК Цена,
|       Сумма КАК Сумма
|   ) КАК Товары,
|   СчетНаОплатуПокупателю.Контрагент КАК Контрагент
|ИЗ
|   Документ.СчетНаОплатуПокупателю КАК СчетНаОплатуПокупателю
|ГДЕ
|   СчетНаОплатуПокупателю.Ссылка = &Ссылка";

Запрос.УстановитьПараметр ("Ссылка", Ссылка);

РезультатЗапроса = Запрос.Выполнить ();

ВыборкаДетальныеЗаписи = РезультатЗапроса.Выбрать ();
```

И в цикле мы будем прописывать заполнение параметров документа.

То есть сейчас мы уже знаем, что в выборке детальные записи, когда начнется цикл, будут содержаться все те поля, которые мы выбрали в запросе.

Перед тем, как заполнять параметры, нужно получить области нашего табличного документа, потому что параметры находятся в областях.

```
ВыборкаДетальныеЗаписи = РезультатЗапроса.Выбрать ();

Шапка = Макет.ПолучитьОбласть ("Шапка");
ТоварыШапка = Макет.ПолучитьОбласть ("ТоварыШапка");
Товары = Макет.ПолучитьОбласть ("Товары");
Подвал = Макет.ПолучитьОбласть ("Подвал");

Пока ВыборкаДетальныеЗаписи.Следующий () Цикл
    // Вставить обработку выборки ВыборкаДетальныеЗаписи
КонецЦикла;
```

Если мы напишем вывод каждой области - то у нас просто выведутся все области с незаполненными в них параметрами.

```
Пока ВыборкаДетальныеЗаписи.Следующий() Цикл
```

```
ТабДок.Вывести(Шапка);  
ТабДок.Вывести(ТоварыШапка);  
ТабДок.Вывести(Товары);  
ТабДок.Вывести(Подвал);
```

```
КонецЦикла;
```

Теперь нам осталось только заполнить параметры.

Благодаря тому, что мы с вами называли параметры в нашем макете точно так же, как они называются в счете на оплату, как они называются в самом документе. Благодаря этому мы можем воспользоваться функцией Заполнить(). И в скобочках мы пишем, откуда мы хотим ее заполнить. А мы хотим ее заполнить из выборки детальные записи.

Для заполнения параметров в каждой области мы пишем вот такую вот строку.

```
Шапка.Параметры.Заполнить(ВыборкаДетальныеЗаписи);  
ТабДок.Вывести(Шапка);
```

Обязательно нужно заполнить все параметры до вывода этой области.

```
Пока ВыборкаДетальныеЗаписи.Следующий() Цикл
```

```
Шапка.Параметры.Заполнить(ВыборкаДетальныеЗаписи);  
ТабДок.Вывести(Шапка);
```

```
ТабДок.Вывести(ТоварыШапка);
```

```
ТабДок.Вывести(Товары);
```

```
Подвал.Параметры.Заполнить(ВыборкаДетальныеЗаписи);  
Подвал.Параметры.СуммаДокументаПрописью = "Здесь будет сумма прописью";  
ТабДок.Вывести(Подвал);
```

```
КонецЦикла;
```

И теперь нам осталось только заполнение табличной части. Как это делать? Во-первых мы уже с вами догадываемся, что нам внутри одного цикла по документам нужно будет организовать еще один цикл по его табличной части.

Во-вторых, нам нужно еще заполнить в ней параметры. Точно таким же образом, как мы заполняли параметры в остальных областях.

```

ТабДок.Вывести(ТоварыШапка);
ВыборкаТовары = ВыборкаДетальныеЗаписи.Товары.Выбрать();
Пока ВыборкаТовары.Следующий() Цикл
    Товары.Параметры.Заполнить(ВыборкаТовары);
    ТабДок.Вывести(Товары);
КонецЦикла;

Подвал.Параметры.Заполнить(ВыборкаДетальныеЗаписи);

```

Сохраняем. Идем пробовать. Перезаходим в документы. Печать.

Внешняя печатная форма. Все отлично. Вот у нас и табличная часть тоже заполнилась всем что нужно.

То есть вот, это наш каркас любой обработки. Мы заполнили все параметры из документов и мы заполнили все параметры из табличных частей. Дальше уже можно заниматься каким-то усложнением, украшением и так далее.

Как, например, мы сейчас заполним дальше, попробуем сумму прописью, просто потому что это очень частая задача.

Мы теперь уже отлично понимаем, что она у нас находится в подвале, поэтому нам нужно ее заполнить перед тем, как написано «вывести подвал».

Воспользуемся функцией, которая нам сделает сумму прописью ЧислоПрописью().

Перейдем в синтаксис-помощник, посмотрим ее синтаксис. Здесь мы можем даже найти пример.

```

Подвал.Параметры.СуммаДокументаПрописью =
ЧислоПрописью(ВыборкаДетальныеЗаписи.СуммаДокумента,"Л = ru_RU; ДП = Истина","рубль, рубля, рублей, м, копейка, копейки, копеек, ж");

```

Строка с рублями для копирования: "рубль, рубля, рублей, м, копейка, копейки, копеек, ж".

Следующая распространенная проблема – это поменять формат даты, если дата эта находится в шаблоне.

```

Шапка.Параметры.Заполнить(ВыборкаДетальныеЗаписи);
Шапка.Параметры.Дата = формат(ВыборкаДетальныеЗаписи.Дата,"ДЛФ=DD");
ТабДок.Вывести(Шапка);

```

И последнее, что я хотела бы рассмотреть с вами в этом уроке, это как вывести на форму то, что вообще никак не относится к документу.

Например, текущее количество бонусов.

Мы понимаем, что нам нужно создать где-то параметр текущее количество бонусов.

Теперь, где нам его заполнять? Мы понимаем, что в области подвал, перед тем, как она вывелась - нам нужно заполнить количество бонусов.

```
Бонусы = 0;

Запрос = Новый Запрос;
Запрос.Текст =
    "ВЫБРАТЬ
    | БонусыКлиентовОстатки.КоличествоБонусовОстаток КАК КоличествоБонусовОстаток
    | ИЗ
    | РегистрНакопления.БонусыКлиентов.Остатки КАК БонусыКлиентовОстатки
    | ГДЕ
    | БонусыКлиентовОстатки.Клиент = &Клиент";

Запрос.УстановитьПараметр("Клиент", ВыборкаДетальныеЗаписи.Контрагент);

РезультатЗапроса = Запрос.Выполнить();

ВыборкаПоБонусам = РезультатЗапроса.Выбрать();

Пока ВыборкаПоБонусам.Следующий() Цикл
    Бонусы = ВыборкаПоБонусам.КоличествоБонусовОстаток;
КонецЦикла;

Подвал.Параметры.КолБонусов = Бонусы;
ТабДок.Вывести(Подвал);
```

Вот таким образом - при помощи запросов, можно в любом месте вводить абсолютно любые данные из базы, которые даже никак не связаны с этим документом. Можно написать запрос в любом месте. Главное – понимать, где его писать. Но теперь мы знаем о существованиях областей и понимаем, что писать заполнение любого параметра нужно перед выводом той области, в которой вы этот параметр разместили.

Источник: <https://t.me/blogpro1c>